In[∘]:=
```
SetDirectory["C:\\drorbn\\AcademicPensieve\\Projects\\Theta"];
```

tex

We start by loading the package \verb$KnotTheory`$ --- it is only needed because it has many specific knots pre-defined:

pdf

In[∘]:=
```
<< KnotTheory`
```

pdf

```
Loading KnotTheory` version of October 29, 2024, 10:29:52.1301.
Read more at http://katlas.org/wiki/KnotTheory.
```

tex

Next we quietly define the commands \verb$Rot$, used to compute rotation numbers, and \verb$PolyPlot$, used to plot polynomials as bar codes and as hexagonal QR codes. Neither is a part of the core of the computation of Θ, so neither is shown; yet we do show some usage examples.

pdf

In[∘]:=
```
(* Rot suppressed *)
```

In[∘]:=
```
Rot[pd_PD] := Module[{n, xs, x, rots, Xp, Xm, front = {1}, k},
   n = Length@pd; rots = Table[0, {2 n}];
   xs = Cases[pd, x_X :> { Xp[x[[4]], x[[1]]]  PositiveQ@x
                           Xm[x[[2]], x[[1]]]  True        ];
   For[k = 1, k ≤ 2 n, ++k,
    If[FreeQ[front, -k],
     front = Flatten@Replace[front, k → (xs /. {
            Xp[k, l_] | Xm[l_, k] :> {l + 1, k + 1, -l},
            Xp[l_, k] | Xm[k, l_] :> (++rots[[l]]; {-l, k + 1, l + 1}),
            _Xp | _Xm :> {}
           }), {1}],
     Cases[front, k | -k] /. {k, -k} :> --rots[[k]];
    ]
   ];
   {xs /. {Xp[i_, j_] :> {+1, i, j}, Xm[i_, j_] :> {-1, i, j}}, rots} ];
Rot[K_] := Rot[PD[K]];
```

pdf

In[∘]:=
```
Rot[Mirror@Knot[3, 1]]
```

Out[∘]=
pdf

```
{{{1, 1, 4}, {1, 3, 6}, {1, 5, 2}}, {0, 0, 0, -1, 0, 0}}
```

tex

We urge the reader to compare the above output with the knot diagram in Section~\ref{ssec:OldFormulas}.

pdf

In[ ]:=
```
(* PolyPlot suppressed *)
```

In[ ]:=
```
PolyPlot1[Δ_] := Module[{crs, m, maxc, minc, s, rect},
  rect = {{0, 0}, {1, 0}, {1, 1}, {0, 1}};
  If[Expand[Δ] === 0, Graphics[],
   m = Max[-Exponent[Δ, T, Min], Exponent[Δ, T, Max]];
   crs = CoefficientRules[T^m Δ, {T}];
   maxc = N@Log@Max@Abs[Last /@ crs];
   minc = N@Log@Min@Select[Abs[Last /@ crs], # > 0 &];
   If[minc == maxc, s[_] = 0, s[c_] := s[c] = (maxc - Log@c) / (maxc - minc)];
   Graphics[crs /. ({x_} → c_) :> {
        Lighter[Which[c == 0, White, c > 0, Red, c < 0, Blue], 0.88 s[Abs@c]],
        Tooltip[Polygon[({x + m - 1/2, 0} + #) & /@ rect], c T^(x-m)]
       }, AspectRatio → Min[1/5, 1/ Sqrt[m + 1]],
     ImagePadding → None, PlotRangePadding → None]
  ]];
Options[PolyPlot2] = {Labeled → False};
PolyPlot2[θ_, OptionsPattern[]] := Module[{crs, m1, m2, maxc, minc, s, hex, p},
  If[Expand[θ] === 0, Graphics[{White, Disk[]}],
   hex = Table[{Cos[α], Sin[α]} / Cos[2 π / 12] / 2, {α, 2 π / 12, 2 π, 2 π / 6}];
   m1 = Max[-Exponent[θ, T1, Min], Exponent[θ, T1, Max]];
   m2 = Max[-Exponent[θ, T2, Min], Exponent[θ, T2, Max]];
   crs = CoefficientRules[T1^m1 T2^m2 θ, {T1, T2}];
   maxc = N@Log@Max@Abs[Last /@ crs];
   minc = N@Log@Min@Select[Abs[Last /@ crs], # > 0 &];
   If[minc == maxc, s[_] = 0, s[c_] := s[c] = (maxc - Log@c) / (maxc - minc)];
   Graphics[{(*{Yellow,Disk[{0,0},1+Cos[2π/12]Norm[{m1,m2}]/Sqrt[2]]},*)
     crs /. ({x1_, x2_} → c_) :> {
        Lighter[Which[c == 0, White, c > 0, Red, c < 0, Blue], 0.88 s[Abs@c]],
        p = {{1, -1/2}, {0, Sqrt[3]/2}}.{x1 - m1, x2 - m2};
        Tooltip[Polygon[(p + #) & /@ hex], c T1^(x1-m1) T2^(x2-m2)],
        If[Not@OptionValue[Labeled], {}, {Black, Text[Style[c T1^(x1-m1) T2^(x2-m2), 30], p]}]
       }
    }, ImagePadding → None, PlotRangePadding → None]
  ]];
PolyPlot[{Δ_, θ_}, opts___Rule] := GraphicsColumn[
  {PolyPlot1[Δ], PolyPlot2[θ, FilterRules[{opts}, Options[PolyPlot2]]]},
  Spacings → Scaled@0.08, ImagePadding → None, PlotRangePadding → None,
  FilterRules[{opts}, Options[GraphicsColumn]]
 ];
```

In[∘]:= $\text{PPDemo} = \text{PolyPlot2}\left[2 + T_1 - T_1 T_2 + T_2 - T_1^{-1} + T_1^{-1} T_2^{-1} - T_2^{-1}, \text{Labeled} \to \text{True}\right]$

Out[∘]=



In[∘]:= Export["PPDemo.pdf", PPDemo]

Out[∘]=

PPDemo.pdf

In[∘]:= $1 / 2 + T_1 - T_1 T_2 + T_2 - T_1^{-1} + T_1^{-1} T_2^{-1} - T_2^{-1}$ // TeXForm

Out[∘]//TeXForm=

```
-T_2 T_1+T_1+T_2-\frac{1}{T_2}+\frac{1}{T_2 T_1}-\frac{1}{T_1}+\frac{1}{2}
```

pdf

In[∘]:= $\text{PolyPlot}\left[\left\{T - 1 + T^{-1},\ T_1 + 2 T_2 - 2 T_1^{-1} T_2^{-1}\right\},\ \text{ImageSize} \to \text{Tiny}\right]$

Out[∘]=

pdf



pdf

In[∘]:= $T_3 = T_1 T_2;$

pdf

In[∘]:= $R_{11}[\{s\_, i\_, j\_\}] =$
$$s \left(1/2 - g_{3ii} + T_2^s g_{1ii} g_{2ji} - g_{1ii} g_{2jj} - \left(T_2^s - 1\right) g_{2ji} g_{3ii} + 2 g_{2jj} g_{3ii} - \left(1 - T_3^s\right) g_{2ji} g_{3ji} - g_{2ii} g_{3jj} - T_2^s g_{2ji} g_{3jj} + g_{1ii} g_{3jj} + \left(\left(T_1^s - 1\right) g_{1ji} \left(T_2^{2s} g_{2ji} - T_2^s g_{2jj} + T_2^s g_{3jj}\right) + \left(T_3^s - 1\right) g_{3ji} \left(1 - T_2^s g_{1ii} - \left(T_1^s - 1\right)\left(T_2^s + 1\right) g_{1ji} + \left(T_2^s - 2\right) g_{2jj} + g_{2ij}\right)\right) / \left(T_2^s - 1\right)\right);$$

pdf

In[∘]:=
```
R12[{s0_, i0_, j0_}, {s1_, i1_, j1_}] = s1 (T1^s0 - 1) (T2^s1 - 1)^-1
   (T3^s1 - 1) g1,j1,i0 g3,j0,i1 ( (T2^s0 g2,i1,i0 - g2,i1,j0) - (T2^s0 g2,j1,i0 - g2,j1,j0));
```

pdf

In[∘]:=
```
Γ1[φ_, k_] = -φ / 2 + φ g3kk;
```

pdf

In[∘]:=
```
Θ[K_] := Module[{Cs, φ, n, A, Δ, G, ev, θ},
   {Cs, φ} = Rot[K]; n = Length[Cs];
   A = IdentityMatrix[2 n + 1];
   Cases[Cs, {s_, i_, j_} :> (A[[{i, j}, {i + 1, j + 1}]] += ( -T^s  T^s - 1
                                                              0     -1    ))];
   Δ = T^((-Total[φ]-Total[Cs[[All,1]]])/2) Det[A];
   G = Inverse[A];
   ev[ℰ_] := Factor[ℰ /. g_{ν_,α_,β_} :> (G[[α, β]] /. T → T_ν)];
   θ = ev[Sum_{k1=1}^n Sum_{k2=1}^n R12[Cs[[k1]], Cs[[k2]]]];
   θ += ev[Sum_{k=1}^n R11[Cs[[k]]]];
   θ += ev[Sum_{k=1}^{2 n} Γ1[φ[[k]], k]];
   Factor@{Δ, (Δ /. T → T1) (Δ /. T → T2) (Δ /. T → T3) θ}
  ];
```

In[∘]:=
```
Expand[Θ[Knot[3, 1]]]
```

KnotTheory: Loading precomputed data in PD4Knots`.

Out[∘]=

$$\left\{-1+\frac{1}{T}+T,\ -\frac{1}{T_1^2}-T_1^2-\frac{1}{T_2^2}-\frac{1}{T_1^2 T_2^2}+\frac{1}{T_1 T_2^2}+\frac{1}{T_1^2 T_2}+\frac{T_1}{T_2}+\frac{T_2}{T_1}+T_1^2 T_2-T_2^2+T_1 T_2^2-T_1^2 T_2^2\right\}$$

In[•]:= **Expand[Θ[Knot[10, 165]]]**

Out[•]=

$$\Big\{-15-\frac{2}{T^2}+\frac{10}{T}+10\,T-2\,T^2,$$

$$-1404-\frac{9}{T_1^4}-\frac{178}{T_1^3}+\frac{607}{T_1^2}+\frac{624}{T_1}+624\,T_1+607\,T_1^2-178\,T_1^3-9\,T_1^4-\frac{9}{T_2^4}-\frac{9}{T_1^4\,T_2^4}+\frac{44}{T_1^3\,T_2^4}-$$

$$\frac{65}{T_1^2\,T_2^4}+\frac{44}{T_1\,T_2^4}-\frac{178}{T_2^3}+\frac{44}{T_1^4\,T_2^3}-\frac{178}{T_1^3\,T_2^3}+\frac{104}{T_1^2\,T_2^3}+\frac{104}{T_1\,T_2^3}+\frac{44\,T_1}{T_2^3}+\frac{607}{T_2^2}-\frac{65}{T_1^4\,T_2^2}+\frac{104}{T_1^3\,T_2^2}+\frac{607}{T_1^2\,T_2^2}-$$

$$\frac{1041}{T_1\,T_2^2}+\frac{104\,T_1}{T_2^2}-\frac{65\,T_1^2}{T_2^2}+\frac{624}{T_2}+\frac{44}{T_1^4\,T_2}+\frac{104}{T_1^3\,T_2}-\frac{1041}{T_1^2\,T_2}+\frac{624}{T_1\,T_2}-\frac{1041\,T_1}{T_2}+\frac{104\,T_1^2}{T_2}+\frac{44\,T_1^3}{T_2}+$$

$$624\,T_2+\frac{44\,T_2}{T_1^3}+\frac{104\,T_2}{T_1^2}-\frac{1041\,T_2}{T_1}+624\,T_1\,T_2-1041\,T_1^2\,T_2+104\,T_1^3\,T_2+44\,T_1^4\,T_2+$$

$$607\,T_2^2-\frac{65\,T_2^2}{T_1^2}+\frac{104\,T_2^2}{T_1}-1041\,T_1\,T_2^2+607\,T_1^2\,T_2^2+104\,T_1^3\,T_2^2-65\,T_1^4\,T_2^2-178\,T_2^3+\frac{44\,T_2^3}{T_1}+$$

$$104\,T_1\,T_2^3+104\,T_1^2\,T_2^3-178\,T_1^3\,T_2^3+44\,T_1^4\,T_2^3-9\,T_2^4+44\,T_1\,T_2^4-65\,T_1^2\,T_2^4+44\,T_1^3\,T_2^4-9\,T_1^4\,T_2^4\Big\}$$

In[•]:= **PolyPlot[Θ[Knot[3, 1]]]**

Out[•]=



In[ ]:=
```
PolyPlot[Θ[Knot["K11n34"]]]
```

KnotTheory: Loading precomputed data in DTCode4KnotsTo11`.

KnotTheory: The GaussCode to PD conversion was written by Siddarth Sankaran at the University of Toronto in the summer of 2005.

Out[ ]=



In[ ]:= `PolyPlot[Θ[Knot["K11n42"]]]`

Out[ ]=



In[ ]:= `PositiveQ /@ PD[TorusKnot[13, 2]]`

Out[ ]=

```
PD[True, True, True, True, True, True, True, True, True, True, True, True, True]
```

In[•]:=
```
PolyPlot[Θ[TorusKnot[13, 2]]]
```

Out[•]=



In[•]:=
```
PolyPlot[Θ[TorusKnot[17, 3]]]
```

Out[•]=



In[ ]:= **PolyPlot[Θ[TorusKnot[13, 5]]]**

Out[ ]=



In[ ]:= **PolyPlot[Θ[TorusKnot[7, 6]]]**

Out[ ]=



In[∘]:=
```
AbsoluteTiming[th = Θ[TorusKnot[22, 7]]]
PolyPlot[th]
```

Out[∘]=

$\left\{4805.79, \left\{\frac{1}{T^{63}}\left(1-T+T^2-T^3+T^4-T^5+T^6\right)\right.\right.$
$\left(1-T+T^7-T^8+T^{11}-T^{12}+T^{14}-T^{15}+T^{18}-T^{19}+ \cdots 18 \cdots +T^{42}-T^{45}+T^{46}-T^{48}+T^{49}-T^{52}+T^{53}-T^{59}+T^{60}\right)$
$\left(1+T-T^7-T^8-T^{11}-T^{12}+T^{14}+T^{15}+T^{18}+T^{19}-T^{21}+T^{23}-T^{25}-T^{26}+T^{28}-T^{30}+T^{32}-T^{34}-T^{35}+T^{37}-T^{39}+T^{41}+\right.$
$\left.\left.\left.T^{42}+T^{45}+T^{46}-T^{48}-T^{49}-T^{52}-T^{53}+T^{59}+T^{60}\right), \frac{63-63\,T_1+63\,T_1^7-63\,T_1^8+ \cdots 30638 \cdots +63\,T_1^{245}\,T_2^{252}-63\,T_1^{251}\,T_2^{252}+63\,T_1^{252}\,T_2^{252}}{T_1^{126}\,T_2^{126}}\right\}\right\}$

Full expression not available (original memory size: 8.1 MB)

Out[∘]=



In[∘]:=
```
tab250 = {{1, 0}}~Join~Table[Θ[K], {K, AllKnots[{3, 10}]}];
```

In[∘]:=
```
g250 = GraphicsGrid[Partition[PolyPlot /@ tab250, 10],
  Spacings → 0, Dividers → All, ImagePadding → None, PlotRangePadding → None]
```

Out[∘]=



In[∘]:=
```
Export["Theta4Rolfsen.pdf", g250]
```

Out[∘]=

```
Theta4Rolfsen.pdf
```

See also https://drorbn.net/AcademicPensieve/Projects/HigherRank/DunfieldKnots/.

In[◦]:=
```
θ[BR[5, {1, 2, 3, -1, 2, 1, 3}]]
```

Out[◦]=

$$\left\{\frac{2-3T+2T^2}{T},\ \frac{1}{T_1^2 T_2^2}\left(9-13T_1+9T_1^2-13T_2+6T_1T_2+6T_1^2T_2-13T_1^3T_2+9T_2^2+6T_1T_2^2-12T_1^2T_2^2+6T_1^3T_2^2+9T_1^4T_2^2-13T_1T_2^3+6T_1^2T_2^3+6T_1^3T_2^3-13T_1^4T_2^3+9T_1^2T_2^4-13T_1^3T_2^4+9T_1^4T_2^4\right)\right\}$$

```
θ[BR[5, {1, 2, 3, -1, 2, 1, 3}]]
```

In[◦]:=
```
PolyPlot@Expand[
  Get["C:\\drorbn\\AcademicPensieve\\Projects\\HigherRank\\DunfieldKnots\\D300.m"][[2]] /.
   {T1 → T₁, T2 → T₂}]
```

Out[◦]=